

---



## CHAPTER - 5 .

---


- 5.1 System Flow Chart
- 5.2 Program Flow Chart
- 5.3 Listing of Computer Program
- 5.4 Computer Output
- 5.5 Analysis


SYSTEM FLOW CHART


MENU PROGRAM




PROGRAM FLOW CHART  
COALE1.BAS  
FOR CREATING DATA FILE










COALB2.DAS  
FOR GENERATING THE SOLUTION OF THE DATA  
AVAILABLE BY COAL PROGRAMMING

































11884

















```

10  REM *****
20  REM ***          GOAL PROGRAMMING PACKAGE          ***
30  REM ***          ~~~~~                               ***
40  REM ***  -----                                     ***
50  REM ***  | Program Develped   ***  Under Guidance  | ***
60  REM ***  |                   ***                   | ***
70  REM ***  |           By       ***           Of      | ***
80  REM ***  |                   ***                   | ***
90  REM ***  |   R. M. Patil     ***   Dr. R. U. Kulkarni | ***
100 REM ***  -----                                     ***
110 REM ***                                               ***
140 REM *****
150 CLS
160 KEY OFF
170 SCREEN 2
180 LOCATE 1,30:PRINT "GOAL PROGRAMMING PACKAGE"
190 LOCATE 2,1
200 REM *** Horizontal Lines ***
210 LINE (35,8) - (600,8)
220 LINE (35,24) - (600,24)
230 LINE (35,40) - (600,40)
240 LINE (35,41) - (600,41)
250 LINE (35,80) - (600,80)
260 LINE (35,110) - (600,110)
270 REM *** Vertical Lines ***
280 LINE (35,8) - (35,110)
290 LINE (70,8) - (70,110)
300 LINE (100,8) - (100,110)
310 LINE (130,8) - (130,110)
320 LINE (132,8) - (132,110)
330 LINE (600,8) - (600,110)

```

```
340 LOCATE 3,35:PRINT "CJ(N+2M) & CJ$(N+2M)"
350 LOCATE 5,40:PRINT "A$(N+2M)"
360 LOCATE 7,6:PRINT "VB"
370 LOCATE 8,6:PRINT "&"
380 LOCATE 9,6:PRINT "VB$"
390 LOCATE 8,10:PRINT "V$"
400 LOCATE 8,15:PRINT "C"
410 LOCATE 8,40:PRINT "A(M,N+2M)"
420 LOCATE 12,10:PRINT "P$"
430 LOCATE 12,14:PRINT "CP"
440 LOCATE 12,40:PRINT "P(K,N+2M)"
450 LOCATE 15,10:PRINT "1. For Creation of data file...  "
460 LOCATE 17,10:PRINT "2. For Processing.....  "
470 LOCATE 19,10:PRINT "3. Exit.....  "
480 LOCATE 21,10:INPUT "  Enter any option.....  ",OP
490 IF OP = 3 THEN CLS:STOP:END
500 IF OP = 1 THEN CHAIN "GOAL01.BAS"
510 IF OP = 2 THEN CHAIN "GOAL02.BAS"
520 GOTO 450
```

```

10  REM *****
20  REM ***          GOAL PROGRAMMING PACKAGE          ***
30  REM ***          ~~~~~                              ***
40  REM ***          -----                              ***
50  REM *** | Program Developed *** Under Guidance | ***
60  REM *** |          ***          | ***
70  REM *** |          By          ***          Of          | ***
80  REM *** |          ***          | ***
90  REM *** |          R. M. Patil ***          Dr. R. U. Kulkarni | ***
100 REM ***          -----                              ***
110 REM ***          ***
120 REM *****
130 REM ***          ***
140 REM *** Working Counters          ***
150 REM ***          ***
160 REM *** ~~~~~***
170 REM *** M - Number Of Constraints          ***
180 REM *** N - Number Of Variables          ***
190 REM *** K - Number Of Priorities          ***
200 REM *** CJ - Array For Cost Of Variables          ***
210 REM *** CJ$- Array For Names Of Cost Of Variables          ***
220 REM *** A$ - Array For Names Of Variables          ***
230 REM *** VB - Array For Cost Of Basic Variables          ***
240 REM *** UB$- Array For Names Of Cost Of Basic Variables          ***
250 REM *** U$ - Array For Basic Variables          ***
260 REM *** C - Array For Initial Values Of Basic Variables          ***
270 REM *** A - Array For Technical Coefficient Of Variables          ***
280 REM *** ~~~~~***
290 REM *** CREATION OF FILE FOR INPUT INFORMATION          ***
300 REM *****

```



```

310 CLS:KEY OFF
320 DIM CJ(50),CJ$(50),A$(50),UB(50),UB$(50),U$(50),C(50),A(50,50)
330 LOCATE 3,25:PRINT "Input File Creation Program"
340 LOCATE 5,10 :PRINT "Enter File Name..... ";
350 INPUT NM$
360 LOCATE 20 ,10 :PRINT "Are You Sure <Y/N/E>.... ";
370 INPUT YN$
380 IF YN$ = "E" OR YN$ = "e" THEN CHAIN "GOAL.BAS"
390 IF YN$ <> "Y" AND YN$ <> "y" THEN 340
400 OPEN "0",#1,NM$
410 CLS
420 LOCATE 3,25:PRINT "Input File Creation Program"
430 GOSUB 1150
440 LOCATE 20,10 :PRINT "Enter No. Of Constraints..(M)"
450 LOCATE 20,50 :INPUT M
460 LOCATE 20,10 :PRINT "Enter No. Of Variables....(N)"
470 LOCATE 20,50 :INPUT N
480 LOCATE 20,10 :PRINT "Enter No. Of Priorities...(K)"
490 LOCATE 20,50 :INPUT K
500 LOCATE 20,10 :PRINT "Are You Sure <Y/N/E>....."
510 LOCATE 20,50 :INPUT YN$
520 IF YN$ = "E" OR YN$ = "e" THEN CLOSE(1):CHAIN "GOAL.BAS"
530 IF YN$ <> "Y" AND YN$ <> "y" THEN 440
540 PRINT #1, "Values Of M. N & K"
550 PRINT #1, M
560 PRINT #1, N
570 PRINT #1, K
580 PRINT #1, "Values Of CJ(N+2M)"
590 FOR J = 1 TO (N + 2 * M)
600     LOCATE 20,10 :PRINT "Enter Cost Of Variables           CJ(";J;") ";

```

```
610     INPUT CJ(J)
620     PRINT #1, CJ(J)
630 NEXT J
640 PRINT #1, "Values Of CJ$(N+2M)"
650 FOR J = 1 TO (N + 2 * M)
660     LOCATE 20,10 :PRINT "Enter Name Of Cost Of Variables           CJ$(";J;") ";
670     INPUT CJ$(J)
680     PRINT #1, CJ$(J)
690 NEXT J
700 PRINT #1, "Values Of A$(N+2M)"
710 FOR J = 1 TO (N + 2 * M)
720     LOCATE 20,10 :PRINT "Enter Names Of Variables             A$(";J;") ";
730     INPUT A$(J)
740     PRINT #1, A$(J)
750 NEXT J
760 PRINT #1, "Values Of VB(M)"
770 FOR I = 1 TO M
780     LOCATE 20,10 :PRINT "Enter Cost Of Basic Variables       VB(";I;") ";
790     INPUT VB(I)
800     PRINT #1, VB(I)
810 NEXT I
820 PRINT #1, "Values Of VB$(M)"
830 FOR I = 1 TO M
840     LOCATE 20,10 :PRINT "Enter Names Of Basic Variables   VB$(";I;") ";
850     INPUT VB$(I)
860     PRINT #1, VB$(I)
870 NEXT I
880 PRINT #1, "Values Of U$(M)"
890 FOR I = 1 TO M
900     LOCATE 20,10 :PRINT "Enter Basic Variables           U$(";I;") ";
```

```
910     INPUT U$(I)
920     PRINT #1, U$(I)
930     NEXT I
940     PRINT #1, "Values Of C(M)"
950     FOR I = 1 TO M
960         LOCATE 20,10 :PRINT "Enter Initial Values Of Basic Variables  C(";I;")  ";
970         INPUT C(I)
980         PRINT #1, C(I)
990     NEXT I
1000    PRINT #1, "Values Of A(M..N+2M)"
1010    FOR I = 1 TO M
1020        FOR J = 1 TO N + 2 * M
1030            LOCATE 20,10 :PRINT "Enter Values Of Tech. Coeff. Of Variables A(";I;",";J;")  ";
1040            INPUT A(I,J)
1050            PRINT #1, A(I,J)
1060        NEXT J
1070    NEXT I
1080    CLOSE(1)
1090    CLS
1100    LOCATE 20 ,10 :PRINT "T H A N K   Y O U"
1110    LOCATE 22,10 :PRINT "S E E   Y O U   A G A I N"
1120    CHAIN "GOAL.BAS"
1130    END
1140    REM *** E N D   O F   P R O G R A M E ***
1150    SCREEN 2
1160    CLS:KEY OFF
1170    LOCATE 1,25:PRINT "Input File Creation Program"
1180    LOCATE 2,1
1190    REM *** Horizontal Lines ***
1200    LINE (35,8) - (600,8)
```

```
1210 LINE (35,24) - (600,24)
1220 LINE (35,40) - (600,40)
1230 LINE (35,41) - (600,41)
1240 LINE (35,80) - (600,80)
1250 LINE (35,110) - (600,110)
1260 REM *** Vertical Lines ***
1270 LINE (35,8) - (35,110)
1280 LINE (70,8) - (70,110)
1290 LINE (100,8) - (100,110)
1300 LINE (130,8) - (130,110)
1310 LINE (132,8) - (132,110)
1320 LINE (600,8) - (600,110)
1330 LOCATE 3,35:PRINT "CJ(N+2M) & CJ$(N+2M)"
1340 LOCATE 5,40:PRINT "A$(N+2M)"
1350 LOCATE 7,6:PRINT "UB"
1360 LOCATE 8,6:PRINT "&"
1370 LOCATE 9,6:PRINT "UB$"
1380 LOCATE 8,10:PRINT "U$"
1390 LOCATE 8,15:PRINT "C"
1400 LOCATE 8,40:PRINT "A(M,N+2M)"
1410 LOCATE 12,10:PRINT "P$"
1420 LOCATE 12,14:PRINT "CP"
1430 LOCATE 12,40:PRINT "P(K,N+2M)"
1440 RETURN
```

```

10  REM *****
20  REM ***           GOAL PROGRAMMING PACKAGE           ***
30  REM ***           ~~~~~                               ***
40  REM ***  -----                                       ***
50  REM ***  | Program Developed   ***   Under Guidance   |   ***
60  REM ***  |                   ***                   |   ***
70  REM ***  |           By           ***           Of           |   ***
80  REM ***  |                   ***                   |   ***
90  REM ***  |   R. M. Patil       ***   Dr. R. U. Kulkarni   |   ***
100 REM ***  -----                                       ***
110 REM ***                                                     ***
120 REM *****
130 REM ***                                                     ***
140 REM *** Working Counters                                     ***
150 REM ***                                                     ***
160 REM *** ~~~~~                                             ***
170 REM *** M   - Number Of Constraints                         ***
180 REM *** N   - Number Of Variables                         ***
190 REM *** K   - Number Of Priorities                       ***
200 REM *** CJ  - Array For Cost Of Variables                 ***
210 REM *** CJ$ - Array For Names Of Cost Of Variables       ***
220 REM *** A$  - Array For Names Of Variables               ***
230 REM *** UB  - Array For Cost Of Basic Variables          ***
240 REM *** UB$ - Array For Names Of Cost Of Basic Variables ***
250 REM *** U$  - Array For Basic Variables                  ***
260 REM *** C   - Array For Initial Values Of Basic Variables ***
270 REM *** A   = Array For Technical Coefficient Of Variables ***
280 REM *** CP  - Array For Value Of The Objective Function at Diff. ***
290 REM ***        Priority Levels                               ***
300 REM *** P$  - Array For Names Of Zj - Cj at Diff. Priority Levels ***

```

```

310 REM *** P      = Array For Values Of Zj = Cj at Diff. Priority Levels***
320 REM *** NA     - New Array For Storing Values Of Tech. Coeff.      ***
330 REM *** E      - Variable used for Creating Priority Names          ***
340 REM *** W      - Variable used for Storing Column Position         ***
350 REM *** U      - Variable used for Storing Row Position            ***
360 REM *** MAX    - Variable used for Storing Maximum Value          ***
370 REM *** MIN    - Variable used for Storing Minimum Value          ***
380 REM *** EQLP   - Variable used for Storing Equal Priority         ***
390 REM *** BIG    - Array To Store Value Of MAX                       ***
400 REM *** SAMEL  - Variable used for Storing Value Of L             ***
410 REM *** SAMEJ  - Variable used for Storing Value Of J             ***
420 REM *** SAMEMAX - Variable used for Storing Value Of MAX         ***
430 REM ***~~~~~**
440 REM *** MAIN PROGRAM FOR GOAL PROGRAMMING                          ***
450 REM *****
460 CLS
470 DIM CJ(50),CJ$(50),A$(50),VB(50),VB$(50),V$(50),C(50),A(50,50)
480 DIM CP(50),P$(50),P(50,50),NA(50,50),BIG(50),SAMEBIG(50)
490 LOCATE 3,25:PRINT "Main Program For Goal Programming"
500 LOCATE 5,10:PRINT "Enter Data File Name..... ";
510 LOCATE 7,10:PRINT "Enter Output Print File Name..... ";
520 LOCATE 5,60:INPUT NI$
530 LOCATE 7,60:INPUT NO$
540 LOCATE 20,10:PRINT "Are You Sure <Y/N/E>.... ";
550 INPUT YN$
560 IF YN$ = "E" OR YN$ = "e" THEN CHAIN "GOAL.BAS"
570 IF YN$ <> "Y" AND YN$ <> "y" THEN 450
580 OPEN "I",#1,NI$
590 OPEN "O",#2,NO$
600 CLS

```

```
610 GOSUB 2910
620 REM *** READ INPUT FILE ***
630 INPUT #1, MESS$
640 INPUT #1, M
650 INPUT #1, N
660 INPUT #1, K
670 INPUT #1, MESS$
680 FOR J = 1 TO (N + 2 * M)
690     INPUT #1, CJ(J)
700 NEXT J
710 INPUT #1, MESS$
720 FOR J = 1 TO (N + 2 * M)
730     INPUT #1, CJ$(J)
740 NEXT J
750 INPUT #1, MESS$
760 FOR J = 1 TO (N + 2 * M)
770     INPUT #1, A$(J)
780 NEXT J
790 INPUT #1, MESS$
800 FOR I = 1 TO M
810     INPUT #1, VB(I)
820 NEXT I
830 INPUT #1, MESS$
840 FOR I = 1 TO M
850     INPUT #1, VB$(I)
860 NEXT I
870 INPUT #1, MESS$
880 FOR I = 1 TO M
890     INPUT #1, V$(I)
900 NEXT I
```

```
910 INPUT #1, MESS$
920 FOR I = 1 TO M
930     INPUT #1, C(I)
940 NEXT I
950 INPUT #1, MESS$
960 FOR I = 1 TO M
970     FOR J = 1 TO (N + 2 * M)
980         INPUT #1, A(I,J)
990     NEXT J
1000 NEXT I
1010 CLOSE(1)
1020 REM             *** READ OVER ***
1030 REM             *** Printing Of Formulated Problem ***
1040 GOSUB 1450
1050 REM             *** Processing & Printing Of ***
1060 REM             ***     Priority Matrix     ***
1070 GOSUB 2030
1080 REM             *** Calculation For Next Tableau ***
1090 FOR L = K TO 1 STEP -1
1100     LET MAX = 0
1110     FOR J = 1 TO (N + 2 * M)
1120         IF P(L,J) <= 0 THEN 1190
1130         IF MAX > P(L,J) THEN 1190
1140         FOR NL = K TO (L + 1) STEP -1
1150             IF P(NL,J) < 0 THEN 1190
1160         NEXT NL
1170         LET MAX = P(L,J)
1180         LET W = J
1190     NEXT J
1200     IF MAX = 0 THEN 1370
```



```

1210 LET EQLP = 0
1220 FOR J = 1 TO (N + 2 * M)
1230 IF P(L,J) <= 0 THEN 1270
1240 IF P(L,J) <> MAX THEN BIG(J) = 0 : GOTO 1270
1250 LET BIG(J) = MAX
1260 LET EQLP = EQLP + 1
1270 NEXT J
1280 IF EQLP = 0 THEN 1370
1290 REM *** Processing For Next Lower Priority ***
1300 REM *** As Values Of MAX Appear More Than 1 ***
1310 IF EQLP > 1 THEN GOSUB 1720
1320 REM *** Processing For Selecting The Key Element ***
1330 GOSUB 2470
1340 GOSUB 1450
1350 GOSUB 2030
1360 GOTO 1100
1370 NEXT L
1380 CLS
1390 LOCATE 20 ,10 :PRINT "T H A N K Y O U"
1400 LOCATE 22,10 :PRINT "S E E Y O U A G A I N"
1410 CLOSE(2)
1420 CHAIN "GOAL.BAS"
1430 END
1440 REM *** END OF PROGRAME ***
1450 REM *****
1460 REM *** SUBROUTINE FOR PRINTING ***
1470 REM *** FORMULATED PROBLEM ***
1480 REM *****
1490 PRINT #2, STRING$(132,"-")
1500 PRINT #2, ,,

```

```

1510 FOR J = 1 TO (N + 2 * M)
1520     PRINT #2, CJ(J);CJ$(J);" ";
1530 NEXT J
1540 PRINT #2,
1550 PRINT #2, STRING$(132,"-")
1560 PRINT #2, ,,
1570 FOR J = 1 TO (N + 2 * M)
1580     PRINT #2, A$(J);" ";
1590 NEXT J
1600 PRINT #2,
1610 PRINT #2, STRING$(132,"=")
1620 FOR I = 1 TO M
1630     PRINT #2, VB(I);VB$(I);" ";V$(I);" ";C(I);" ",
1640     FOR J = 1 TO (N + 2 * M)
1650         PRINT #2, A(I,J);" ";
1660     NEXT J
1670     PRINT #2,
1680 NEXT I
1690 PRINT #2, STRING$(132,"-")
1700 RETURN
1710 REM *** END OF PROGRAM ***
1720 REM *****
1730 REM *** SUBROUTINE FOR SAME PRIORITY ***
1740 REM *****
1750 IF K = 1 THEN 2010
1760 LET SAMEMAX = 0
1770 LET SAMEP = 0
1780 FOR SAMEL = K - 1 TO 1 STEP -1
1790     FOR SAMEJ = 1 TO (N + 2 * M)
1800         IF BIG(SAMEJ) <> MAX THEN 1840

```

```

1810     IF SAMEMAX >= P(SAMEL, SAMEJ) THEN 1840
1820     LET SAMEMAX = P(SAMEL, SAMEJ)
1830     LET SAMEW = SAMEJ
1840     NEXT SAMEJ
1850     IF SAMEMAX = 0 THEN 1970
1860     FOR SAMEJ = 1 TO (N + 2 * M)
1870         IF P(SAMEL, SAMEJ) <= 0 THEN 1930
1880         IF P(SAMEL, SAMEJ) = SAMEMAX THEN 1910
1890         LET BIG(SAMEJ) = 0
1900         GOTO 1930
1910         LET BIG(SAMEJ) = SAMEMAX
1920         LET SAMEP = SAMEP + 1
1930     NEXT SAMEJ
1940     LET MAX = SAMEMAX
1950     IF SAMEP > 1 THEN 1970
1960     GOTO 1980
1970 NEXT SAMEL
1980 IF SAMEP = 0 THEN 2010
1990 LET MAX = SAMEMAX
2000 LET W = SAMEW
2010 RETURN
2020 REM *** END OF PROGRAME ***
2030 REM *****
2040 REM *** SUBROUTINE FOR PRINTING ***
2050 REM *** PRIORITY TABLEAU ***
2060 REM *****
2070 REM *** PROCESSING STARTS ***
2080 E = 0
2090 REM *** Initialize Values Of P(K, N+2M) ***
2100 FOR L = K TO 1 STEP -1

```

```

2110 LET CP(L) = 0
2120 LET E = E + 1
2130 LET P$(L) = "P"+MID$(STR$(E),2,2)
2140 FOR J = 1 TO (N + 2 * M)
2150 LET P(L,J) = 0
2160 NEXT J
2170 NEXT L
2180 REM *** Calculation Of CP(K) & P(K, N+2M) ***
2190 FOR L = K TO 1 STEP -1
2200 FOR I = 1 TO M
2210 IF VB$(I) <> P$(L) THEN 2260
2220 LET CP(L) = CP(L) + C(I) * VB(I)
2230 FOR J = 1 TO (N + 2 * M)
2240 LET P(L,J) = P(L,J) + VB(I) * A(I,J)
2250 NEXT J
2260 NEXT I
2270 NEXT L
2280 REM *** Calculation Of P(K, N+2M) i.e. Zj - Cj ***
2290 FOR L = 1 TO K
2300 FOR J = 1 TO (N + 2 * M)
2310 IF P$(L) = CJ$(J) THEN P(L,J) = P(L,J) - CJ(J)
2320 NEXT J
2330 NEXT L
2340 REM *** Print Priority Matrix ***
2350 FOR L = 1 TO K
2360 PRINT #2, " ";P$(L),CP(L),
2370 FOR J = 1 TO (N + 2 * M)
2380 PRINT #2, P(L,J);" ";
2390 NEXT J
2400 PRINT #2,

```

```

2410 NEXT L
2420 PRINT #2,
2430 PRINT #2, STRING$(132,"-")
2440 PRINT #2, : PRINT #2, : PRINT #2, : PRINT #2, : PRINT #2,
2450 RETURN
2460 REM *** END OF PROGRAME ***
2470 REM *****
2480 REM *** SUBROUTINE FOR SELECTING ***
2490 REM *** THE KEY ELEMENT ***
2500 REM *****
2510 LET MIN = 999999!
2520 FOR I = 1 TO M
2530 IF A(I,W) <= 0 THEN 2550
2540 IF MIN > C(I) / A(I,W) THEN LET MIN = C(I) / A(I,W) : LET U = I
2550 NEXT I
2560 IF MIN = 999999! THEN PRINT #2, "Unbounded Solution":GOTO 1390
2570 REM *** Transformation Of New Values ***
2580 REM *** Of Next Tableau Into Original ***
2590 REM *** Variables ***
2600 LET VB(U) = CJ(W)
2610 LET VB$(U) = CJ$(W)
2620 LET V$(U) = A$(W)
2630 LET C(U) = C(U) / A(U,W)
2640 FOR J = 1 TO (N + 2 * M)
2650 IF J = W THEN 2670
2660 LET A(U,J) = A(U,J) / A(U,W)
2670 NEXT J
2680 LET A(U,W) = 1
2690 FOR I = 1 TO M
2700 FOR J = 1 TO (N + 2 * M)

```

```
2710     LET NA(I,J) = A(I,J)
2720     NEXT J
2730     NEXT I
2740     FOR I = 1 TO M
2750     IF I = U THEN 2800
2760     LET C(I) = C(I) - C(U) * A(I,W)
2770     FOR J = 1 TO (N + 2 * M)
2780     LET NA(I,J) = NA(I,J) - A(U,J) * A(I,W)
2790     NEXT J
2800     NEXT I
2810     FOR I = 1 TO M
2820     FOR J = 1 TO (N + 2 * M)
2830     LET A(I,J) = NA(I,J)
2840     NEXT J
2850     NEXT I
2860     LOCATE 20,30:PRINT RRP;" Iteration is Over"
2870     LOCATE 22,30:INPUT "Strick Any Key To Continue ",RMP
2880     LET RRP = RRP + 1
2890     RETURN
2900     REM *** E N D   O F   P R O G R A M E ***
2910     REM *** Subroutine for screen ***
2920     CLS
2930     KEY OFF
2940     SCREEN 2
2950     LOCATE 1,25:PRINT "Main Program For Goal Programming"
2960     LOCATE 2,1
2970     REM *** Horizontal Lines ***
2980     LINE (35,8) - (600,8)
2990     LINE (35,24) - (600,24)
3000     LINE (35,40) - (600,40)
```

```
3010 LINE (35,41) - (600,41)
3020 LINE (35,80) - (600,80)
3030 LINE (35,110) - (600,110)
3040 REM *** Vertical Lines ***
3050 LINE (35,8) - (35,110)
3060 LINE (70,8) - (70,110)
3070 LINE (100,8) - (100,110)
3080 LINE (130,8) - (130,110)
3090 LINE (132,8) - (132,110)
3100 LINE (600,8) - (600,110)
3110 LOCATE 3,35:PRINT "CJ(N+2M) & CJ$(N+2M)"
3120 LOCATE 5,40:PRINT "A$(N+2M)"
3130 LOCATE 7,6:PRINT "UB"
3140 LOCATE 8,6:PRINT "&"
3150 LOCATE 9,6:PRINT "UB$"
3160 LOCATE 8,10:PRINT "U$"
3170 LOCATE 8,15:PRINT "C"
3180 LOCATE 8,40:PRINT "A(M,N+2M)"
3190 LOCATE 12,10:PRINT "P$"
3200 LOCATE 12,14:PRINT "CP"
3210 LOCATE 12,40:PRINT "P(K,N+2M)"
3220 RETURN
```

### 5.4 Computer Output :-

The following table represents the data regarding first sector concerning Control Points Bhawani Mandap and Shivaji Maidan.

T A B L E - I

I. B. M. / S. P.	!Passengers!	!Trips!	Km.	Revenue	!Time !
1. Sugar mill	1505	30	7.00	2333.00	25
2. Apate Nagar	1386	42	4.50	1682.75	15
3. Katyayani	659	18	9.50	384.50	30
4. Salokhe Nagar	1217	25	4.60	1877.75	20
5. Vashi	825	28	9.50	1498.00	30
6. University Colony	1179	32	6.00	1679.25	25
7. University	865	30	6.00	1259.00	20
8. Rajarampuri	994	40	4.00	1450.00	15
9. K.I.T.	860	44	9.50	1207.75	30
10. Samrat Nagar	759	32	4.50	1258.75	20
11. Jarag Nagar	1297	38	4.60	1703.75	20
12. Nana Patil Nagar	1386	42	4.50	1682.75	20
1. Kadamwadi	1467	31	9.10	1955.00	40
2. Market yard	775	26	5.50	575.00	20
3. University	792	28	6.00	1000.25	20
4. Apte Nagar	1389	42	4.00	1610.50	15
	17355	528	98.80	23158.00	365

It is observed from the above table that --

- a) Total number of passengers travelled per day - 17,355.00
- b) Total number of trips per day - 528.00
- c) Total number of Kilometer travelled per day - 98.80
- d) Total number of Revenue earned per day - 23,158.00
- e) Total number of Time taken in minutes per day - 365.00



The following table represents the data regarding first sector concerning Control Points Gangavesh and Shahu Maidan.

TABLE - II

II. G. / S. M.	Passengers	Trips	Km.	Revenue	Time
1. Kuditre	1082	21	14.50	2301.00	45
2. Koge	926	22	11.80	1800.00	40
3. Wakare	1400	24	10.50	2711.00	40
4. Hanumantwadi	1537	32	9.90	2665.00	25
5. Padali	616	18	7.80	1164.75	25
6. Shindewadi	1002	22	11.40	1572.75	40
7. Shinganapur	1156	32	7.00	1941.50	25
8. University	601	30	6.00	885.00	25
9. Bavada	1237	22	6.80	1961.00	25
1. Mudshingi	832	22	9.80	1345.50	35
2. Nerle Tamgaon	1198	20	13.20	2325.75	40
3. Kaneri Math	444	14	15.40	824.25	45
4. More wadi	1250	32	6.00	1989.25	25
5. Uchgaon	1014	32	7.30	1525.00	25
6. Lonar Vasahat	286	18	6.70	397.00	25
7. Rajendra Nagar	903	30	6.00	1227.25	25
	15484	391	150.10	26636.00	510

It is observed from the above table that --

- a) Total number of passengers travelled per day - 15,484.00
- b) Total number of trips per day - 391.00
- c) Total number of Kilometer travelled per day - 150.10
- d) Total number of Revenue earned per day - 26,636.00
- e) Total number of Time taken in minutes per day - 510.00

The following table represents the data regarding first sector concerning Control Points Maharana Pratap Chowk and Sonya Maruti Chowk.

TABLE - III

III. M. P. C. / S. M. C.	Passengers	Trips	Km.	Revenue	Time
1. Shiye	818	16	17.50	1628.75	50
2. Chinchwad	1168	24	11.00	2119.25	35
3. Waliwade	1599	26	11.00	3109.50	35
4. Koyana Colony	1059	22	11.00	1895.75	35
5. Chokak	843	18	18.00	1731.75	50
1. Jathar Wadi	788	18	13.00	1612.25	40
2. Porle	888	22	14.10	1952.50	40
3. Nitawade	750	26	11.20	1461.00	35
	7913	172	106.80	15510.75	320

It is observed from the above table that --

- a) Total number of passengers travelled per day - 7,913.00
- b) Total number of trips per day - 172.00
- c) Total number of Kilometer travelled per day - 106.80
- d) Total number of Revenue earned per day - 15,510.75
- e) Total number of Time taken in minutes per day - 320.00

Thus, it is observed that the total data of all the sectors is as follows --

*Table - 9*

a) Total number of passengers travelled per day	- 40,752.00
b) Total number of trips per day	- 1,091.00
c) Total number of Kilometer travelled per day	- 355.70
d) Total number of Revenue earned per day	- 65,304.75
e) Total number of Time taken in minutes per day	- 1,195.00

#### **Formulation Of The Problem :-**

Let,  $x_1$  represents 1st sector i.e. Bhavani Mandap and Shivaji Putla based on sectoral route length.

$x_2$  represents 2nd sector i.e. Gangavesh and Shahu Maidan based on sectoral route length.

$x_3$  represents 3rd sector i.e. Maharana Pratap Chowk and Sonya Maruti Chowk based on sectoral route length.

$d_1$  represents positive deviation from goal 1

$d_2$  represents positive deviation from goal 2

$d_3$  represents positive deviation from goal 3

$d_1$  represents negative deviation from goal 1

$d_2$  represents negative deviation from goal 2

$d_3$  represents negative deviation from goal 3

Thus, the objective function (Z) becomes --

$$\text{Minimize, } Z = P_1(d_1^- + d_3^-) + P_2 d_2 + P_3 d_3$$

Subject to Goal Constraints,

Occupancy Constraint :-

$$32.87x_1 + 39.60x_2 + 46.01x_3 + d_1^- - d_1^+ = 1494.12$$

Revenue Constraint :-

$$43.86x_1 + 68.12x_2 + 90.18x_3 + d_2^- - d_2^+ = 2394.31$$

Bus Availability Constraint :-

$$1.00x_1 + 1.00x_2 + 1.00x_3 + d_3^- - d_3^+ = 40.00$$

## GOAL PROGRAMMING SOLUTION

0th Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	1494.120	32.87	39.60	46.01	1	0	0	0	0	0
1 P2	d2-	2394.310	43.86	68.12	90.18	0	1	0	0	0	0
1 P3	d3-	40.000	1.00	1.00	1.00	0	0	1	0	0	-1
	P3	40.000	1	1	1	0	0	0	0	0	-1
	P2	2394.310	43.86	68.12	90.18	0	0	0	0	0	0
	P1	1494.120	32.87	39.60	46.01	0	0	0	0	0	-1

1st Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	272.539	10.49	4.85	0	1	-0.51	0	0	0	0
$\emptyset -$	.x3	26.550	0.49	0.76	1	0	0.01	0	0	0	0
1 P3	d3-	13.450	0.51	0.24	0	0	-0.01	1	0	0	-1
	P3	13.450	0.51	0.24	0	0	-0.01	0	0	0	-1
	P2	0.000	0.00	0.00	0	0	-1.00	0	0	0	0
	P1	272.539	10.49	4.85	0	0	-0.51	0	0	0	-1

## 2nd Iteration

			$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
			.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x1	25.975	1	0.46	0	0.10	-0.05	0	0	0	0
$\emptyset -$	.x3	13.917	0	0.53	1	-0.05	0.03	0	0	0	0
1 P3	d3-	0.108	0	0.01	0	-0.05	0.01	1	0	0	-1
	P3	0.108	0	0.01	0	-0.05	0.01	0	0	0	-1
	P2	0	0	0	0	0	-1	0	0	0	0
	P1	0	0	0	0	-1	0	0	0	0	-1

## 3rd Iteration

			$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
			.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x1	19.266	1	0	0	3.13	-0.91	-62.04	0	0	62.04
$\emptyset -$	.x3	6.206	0	0	1	3.44	-0.96	-71.31	0	0	71.31
$\emptyset -$	.x2	14.527	0	1	0	-6.58	1.87	134.35	0	0	-134.35
	P3	0	0	0	0	0	0	-1	0	0	0
	P2	0	0	0	0	0	-1	0	0	0	0
	P1	0	0	0	0	-1	0	0	0	0	-1

Hence, the optimal solution obtained is --

$$x_1 = 19$$

$$x_2 = 15$$

$$x_3 = 6$$

So in order to maximize occupancy, revenue the Kolhapur Municipal Transport should run 19 buses to sector 1 i.e. Bhawani Mandap and Shivaji Putla, 15 buses to sector 2 i.e. Gangavesh and Shahu Maidan and 6 buses to sector 3 i.e. Maharana Pratap Chowk and Sonya Maruti Chowk.

The following table represents the data regarding first sector concerning Control Points Bhawani Mandap and Shivaji Putla.

T A B L E - I V

	1	2	3	4	5
I. BHAVANI MANDAP	Passengers	Trips	Km.	Revenue	Time
1. Sugar mill	1505	30	7.00	2333.00	25
2. Apate Nagar	1386	42	4.50	1682.75	15
3. Katyayani	659	18	9.50	384.50	30
4. Salokhe Nagar	1217	25	4.60	1877.75	20
5. Vashi	825	28	9.50	1498.00	30
6. University Colony	1179	32	6.00	1679.25	25
7. University	865	30	6.00	1259.00	20
8. Rajarampuri	994	40	4.00	1450.00	15
9. K.I.T.	860	44	9.50	1207.75	30
10. Samrat Nagar	759	32	4.50	1258.75	20
11. Jarag Nagar	1297	38	4.60	1703.75	20
12. Nana Patil Nagar	1386	42	4.50	1682.75	20
	12932	401	74.20	18017.25	270
II. SHIVAJI PUTLA	Passengers	Trips	Km.	Revenue	Time
1. Kadamwadi	1467	31	9.10	1955.00	40
2. Market yard	775	26	5.50	575.00	20
3. University	792	28	6.00	1000.25	20
4. Apte Nagar	1389	42	4.00	1610.50	15
	4423	127	24.60	5140.75	95
TOTAL :-	17355	528	98.80	23158.00	365



It is observed from the above table that --

- a) Total number of passengers travelled per day - 17,355.00
- b) Total number of trips per day - 528.00
- c) Total number of Kilometer travelled per day - 98.80
- d) Total number of Revenue earned per day - 23,158.00
- e) Total number of Time taken in minutes per day - 365.00

**Formulation Of The Problem For The 1st Sector :-**

Let,  $x_1$  represents Buses to be run from Bhavani Mandap

$x_2$  represents Buses to be run from Shivaji Putla

$d_1^-$  represents positive deviation from goal 1

$d_2^-$  represents positive deviation from goal 2

$d_1^+$  represents negative deviation from goal 1

$d_2^+$  represents negative deviation from goal 2

Thus, the objective function (Z) becomes --

$$\text{Minimize, } Z = P_1(d_1^- + d_1^+) + P_2(d_2^- + d_2^+)$$

Subject to Goal Constraints,

Occupancy Constraint :-

$$32.25x_1 + 34.83x_2 + d_1^- - d_1^+ = 624.52$$

Revenue Constraint :-

$$44.93x_1 + 40.48x_2 + d_2^- - d_2^+ = 833.34$$

Bus Availability Constraint :-

$$1.00x_1 + 1.00x_2 + d_3^- - d_3^+ = 19.00$$

## GOAL PROGRAMMING SOLUTION

## 0th Iteration

		$\emptyset$ —	$\emptyset$ —	1 P1	1 P2	1 P3	$\emptyset$ —	$\emptyset$ —	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
1 P1	d1-	624.520	32.25	34.38	1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P2	d2-	833.340	44.93	40.48	$\emptyset$	1	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	19.000	1.00	1.00	$\emptyset$	$\emptyset$	1	$\emptyset$	-1
	F3	19.000	1	1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1
	F2	833.340	44.93	40.48	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
	F1	624.520	32.25	34.38	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1

## 1st Iteration

		$\emptyset$ —	$\emptyset$ —	1 P1	1 P2	1 P3	$\emptyset$ —	$\emptyset$ —	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset$ —	.x2	18.165	0.938	1	0.029	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P2	d2-	98.012	6.958	$\emptyset$	-1.177	1	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	0.835	0.062	$\emptyset$	-0.029	$\emptyset$	1	$\emptyset$	-1
	F3	0.835	0.062	$\emptyset$	-0.029	$\emptyset$	$\emptyset$	$\emptyset$	-1
	F2	98.012	6.958	$\emptyset$	-1.177	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
	F1	0.000	0.000	$\emptyset$	-1.000	$\emptyset$	$\emptyset$	$\emptyset$	-1

## 2nd Iteration

		$\emptyset$ —	$\emptyset$ —	1 P1	1 P2	1 P3	$\emptyset$ —	$\emptyset$ —	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset$ —	.x2	5.526	$\emptyset$	1	.469	$\emptyset$	-15.141	$\emptyset$	15.141
1 P2	d2-	4.260	$\emptyset$	$\emptyset$	2.009	1	-112.307	$\emptyset$	112.307
$\emptyset$ —	.x1	13.474	1	$\emptyset$	-.469	$\emptyset$	16.141	$\emptyset$	-16.141
	F3	0.000	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$
	F2	4.260	$\emptyset$	$\emptyset$	2.009	$\emptyset$	-112.307	$\emptyset$	112.307
	F1	0.000	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$	-1

Hence, the optimal solution obtained is --

$$x_1 = 13$$

$$x_2 = 6$$

So in order to maximize occupancy, revenue the Kolhapur Municipal Transport should run 13 buses to Control Point 1 i.e. Bhawani Mandap and 6 buses to Control Point 2 i.e. Shivaji Putla.

The following table represents the data regarding second sector concerning Control Points Gangavesh and Shahu Maidan.

T A B L E - V

	1	2	3	4	5
I. GANGAVESH	Passengers	Trips	Km.	Revenue	Time
1. Kuditre	1082	21	14.50	2301.00	45
2. Koge	926	22	11.80	1800.00	40
3. Wakare	1400	24	10.50	2711.00	40
4. Hanumantwadi	1537	32	9.90	2665.00	25
5. Padali	616	18	7.80	1164.75	25
6. Shindewadi	1002	22	11.40	1572.75	40
7. Shinganapur	1156	32	7.00	1941.50	25
8. University	601	30	6.00	885.00	25
9. Bavada	1237	22	6.80	1961.00	25
	9557	223	85.70	17002.00	290
II. SHAHU MAIDAN	Passengers	Trips	Km.	Revenue	Time
1. Mudshingi	832	22	9.80	1345.50	35
2. Nerle Tamgaon	1198	20	13.20	2325.75	40
3. Kaneri Math	444	14	15.40	824.25	45
4. More wadi	1250	32	6.00	1989.25	25
5. Uchgaon	1014	32	7.30	1525.00	25
6. Lonar Vasahat	286	18	6.70	397.00	25
7. Rajendra Nagar	903	30	6.00	1227.25	25
	5927	168	64.40	9634.00	220
<b>TOTAL :-</b>	<b>15484</b>	<b>391</b>	<b>150.10</b>	<b>26636.00</b>	<b>510</b>

It is observed from the above table that --

- a) Total number of passengers travelled per day - 15,484.00
- b) Total number of trips per day - 391.00
- c) Total number of Kilometer travelled per day - 150.10
- d) Total number of Revenue earned per day - 26,636.00
- e) Total number of Time taken in minutes per day - 510.00

**Formulation Of The Problem For The 2nd Sector :-**

Let,  $x_1$  represents Buses to be run from Gangavesh  
 $x_2$  represents Buses to be run from Shahu Maidan  
 $d_1^-$  represents positive deviation from goal 1  
 $d_2^-$  represents positive deviation from goal 2  
 $d_1^+$  represents negative deviation from goal 1  
 $d_2^+$  represents negative deviation from goal 2

Thus, the objective function (Z) becomes --

$$\text{Minimize, } Z = P_1(d_1^- + d_2^-) + P_2(d_1^+ + d_2^+)$$

Subject to Goal Constraints,

Occupancy Constraint :-

$$42.86x_1 + 35.28x_2 + d_1^- - d_1^+ = 594.02$$

Revenue Constraint :-

$$76.24x_1 + 57.35x_2 + d_2^- - d_2^+ = 1021.84$$

Bus Availability Constraint :-

$$1.00x_1 + 1.00x_2 + d_3^- - d_3^+ = 15.00$$

## GOAL PROGRAMMING SOLUTION

## 0th Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	594.020	42.86	35.28	1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P2	d2-	1021.840	76.24	57.35	$\emptyset$	1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	15.000	1	1	$\emptyset$	$\emptyset$	1	$\emptyset$	$\emptyset$	-1
	P3	15.000	1	1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1
	P2	1021.840	76.24	57.35	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
	P1	594.020	42.86	35.28	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1

## 1st Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	19.570	$\emptyset$	3.039	1	-0.562	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
$\emptyset -$	.x1	13.403	1	0.752	$\emptyset$	0.013	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	1.597	$\emptyset$	0.248	$\emptyset$	-0.013	1	$\emptyset$	$\emptyset$	-1
	P3	1.597	$\emptyset$	0.248	$\emptyset$	-0.013	$\emptyset$	$\emptyset$	$\emptyset$	-1
	P2	0.000	$\emptyset$	0.000	$\emptyset$	-1.000	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
	P1	19.570	$\emptyset$	3.039	$\emptyset$	-0.562	$\emptyset$	$\emptyset$	$\emptyset$	-1

## 2nd Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x2	6.439	$\emptyset$	1	0.329	-0.185	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
$\emptyset -$	.x1	8.560	1	$\emptyset$	-0.247	0.152	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	0.002	$\emptyset$	$\emptyset$	-0.032	0.033	1	$\emptyset$	$\emptyset$	-1
	P3	0.002	$\emptyset$	$\emptyset$	-0.032	0.032	$\emptyset$	$\emptyset$	$\emptyset$	-1
	P2	0.000	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$
	P1	0.000	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1

Hence, the optimal solution obtained is --

$$x_1 = 9$$

$$x_2 = 6$$

So in order to maximize occupancy, revenue the Kolhapur Municipal Transport should run 9 buses to Control Point 3 i.e. Gangavesh and 6 buses to Control Point 4 i.e. Shahu Maidan.

The following table represents the data regarding third sector concerning Control Points Maharana Pratap Chowk and Sonya Maruti Chowk.

T A B L E - V I

	1	2	3	4	5
I. MAHARANA PRATAP CHOWK	(Passengers)	(Trips)	Km.	Revenue	(Time)
1. Shiye	818	16	17.50	1628.75	50
2. Chinchwad	1168	24	11.00	2119.25	35
3. Waliwade	1599	26	11.00	3109.50	35
4. Koyana Colony	1059	22	11.00	1895.75	35
5. Chokak	843	18	18.00	1731.75	50
	5487	106	68.50	10485.00	205
II. SONYA MARUTI CHOWK	(Passengers)	(Trips)	Km.	Revenue	(Time)
1. Jathar Wadi	788	18	13.00	1612.25	40
2. Porle	888	22	14.10	1952.50	40
3. Nitawade	750	26	11.20	1461.00	35
	2426	66	38.30	5025.75	115
TOTAL :-	7913	172	106.80	15510.75	320



It is observed from the above table that --

- a) Total number of passengers travelled per day - 7,913.00
- b) Total number of trips per day - 172.00
- c) Total number of Kilometer travelled per day - 106.80
- d) Total number of Revenue earned per day - 15,510.75
- e) Total number of Time taken in minutes per day - 320.00

#### Formulation Of The Problem For The 3rd Sector :-

Let,  $x_1$  represents Buses to be run from Maharana Pratap  
Chowk

$x_2$  represents Buses to be run from Sonya Maruti  
Chowk

$d_1$  represents positive deviation from goal 1

$d_2$  represents positive deviation from goal 2

$d_1$  represents negative deviation from goal 1

$d_2$  represents negative deviation from goal 2

Thus, the objective function (Z) becomes --

$$\text{Minimize, } Z = P_1(d_1 + d_2) + P_2 d_2$$

Subject to Goal Constraints,

Occupancy Constraint :-

$$51.76x_1 + 36.76x_2 + d_1^- - d_1^+ = 276.03$$

Revenue Constraint :-

$$98.92x_1 + 76.15x_2 + d_2^- - d_2^+ = 541.07$$

Bus Availability Constraint :-

$$1.00x_1 + 1.00x_2 + d_3^- - d_3^+ = 6.00$$

## GOAL PROGRAMMING SOLUTION

## 0th Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
1 P1	d1-	276.030	51.76	36.76	1	0	0	0	0
1 P2	d2-	541.070	98.92	76.15	0	1	0	0	0
1 P3	d3-	6.000	1	1	0	0	1	0	-1
	P3	6.000	1	1	0	0	0	0	-1
	P2	541.070	98.92	76.15	0	0	0	0	0
	P1	276.030	51.76	36.76	0	0	0	0	-1

## 1st Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x1	5.333	1	0.710	0.019	0	0	0	0
1 P2	d2-	13.541	0	5.897	-1.911	1	0	0	0
1 P3	d3-	0.667	0	0.290	-0.019	0	1	0	-1
	P3	0.667	0	0.290	-0.019	0	0	0	-1
	P2	13.541	0	5.897	-1.911	0	0	0	0
	P1	0.000	0	0	-1	0	0	0	-1

## 2nd Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x1	3.702	1	0	0.249	-0.120	0	0	0
$\emptyset -$	.x2	2.296	0	1	-0.324	0.170	0	0	0
1 P3	d3-	0.002	0	0	0.075	-0.049	1	0	-1
	P3	0.002	0	0	0.075	-0.049	0	0	-1
	P2	0.000	0	0	0	-1	0	0	0
	P1	0.000	0	0	-1	0	0	0	-1

Hence, the optimal solution obtained is --

$$x_1 = 4$$

$$x_2 = 2$$

So in order to maximize occupancy, revenue the Kolhapur Municipal Transport should run 4 buses to Control Point 5 i.e. Maharana Pratap Chowk and 2 buses to Control Point 6 i.e. Sonya Maruti Chowk.

### Formulation Of The Problem

#### For Revenue Maximization :-

The same problem may be run even with Revenue as first priority and Occupancy as second priority. Thus, the formulation is as follows --

Let,  $x_1$  represents 1st sector i.e. Bhavani Mandap and Shivaji Putla based on sectoral route length.

$x_2$  represents 2nd sector i.e. Gangavesh and Shahu Maidan based on sectoral route length.

$x_3$  represents 3rd sector i.e. Maharana Pratap Chowk and Sonya Maruti Chowk based on sectoral route length.

$d_1$  represents positive deviation from goal 1

$d_2$  represents positive deviation from goal 2

$d_3$  represents positive deviation from goal 3

$d_1$  represents negative deviation from goal 1

$d_2$  represents negative deviation from goal 2

$d_3$  represents negative deviation from goal 3

Thus, the objective function (Z) becomes --

$$\text{Minimize, } Z = P_1(d_1 + d_3) + P_2 d_2 + P_3 d_3$$

Subject to Goal Constraints,

Revenue Constraint :-

$$43.86x_1 + 68.12x_2 + 90.18x_3 + d_2^- - d_2^+ = 2394.31$$

Occupancy Constraint :-

$$32.87x_1 + 39.60x_2 + 46.01x_3 + d_1^- - d_1^+ = 1494.12$$

Bus Availability Constraint :-

$$1.00x_1 + 1.00x_2 + 1.00x_3 + d_3^- - d_3^+ = 40.00$$

## GOAL PROGRAMMING SOLUTION

0th Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 F1	1 F2	1 F3	$\emptyset -$	$\emptyset -$	1 F1	
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+	
1 F1	d1-	2394.310	43.86	68.12	90.18	0	1	0	0	0	0
1 F2	d2-	1494.120	32.87	39.60	46.01	1	0	0	0	0	0
1 F3	d3-	40.000	1	1	1	0	0	1	0	0	-1
	F3	40.000	1	1	1	0	0	0	0	0	-1
	F2	1494.120	32.87	39.6	46.01	1	-1	0	0	0	0
	F1	2394.310	43.86	68.12	90.18	-1	1	0	0	0	-1

1st Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 F1	1 F2	1 F3	$\emptyset -$	$\emptyset -$	1 F1	
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x3	26.550	0.49	0.76	1	0	0.01	0	0	0	0
1 F2	d2-	272.539	10.49	4.85	0	1	-0.51	0	0	0	0
1 F3	d3-	13.450	0.51	0.24	0	0	-0.01	1	0	0	-1
	F3	13.450	0.51	0.24	0	0	-0.01	0	0	0	-1
	F2	272.539	10.49	4.85	0	1	-1.51	0	0	0	0
	F1	0	0	0	0	-1	0	0	0	0	-1

## 2nd Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x3	13.917	$\emptyset$	0.531	1	-0.05	0.03	$\emptyset$	$\emptyset$	$\emptyset$
$\emptyset -$	.x1	25.975	1	0.462	$\emptyset$	0.10	-0.05	$\emptyset$	$\emptyset$	$\emptyset$
1 P3	d3-	0.108	$\emptyset$	0.007	$\emptyset$	-0.05	0.01	1	$\emptyset$	$\emptyset$
	P3	0.108	$\emptyset$	0.01	$\emptyset$	-0.05	0.01	$\emptyset$	$\emptyset$	$\emptyset$
	P2	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$
	P1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$	-1

## 3rd Iteration

		$\emptyset -$	$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1
		.x1	.x2	.x3	d1-	d2-	d3-	d1+	d2+	d3+
$\emptyset -$	.x3	6.206	$\emptyset$	$\emptyset$	1	3.44	-0.96	-71.31	$\emptyset$	$\emptyset$
$\emptyset -$	.x1	19.266	1	$\emptyset$	$\emptyset$	3.13	-0.91	-62.04	$\emptyset$	$\emptyset$
$\emptyset -$	.x2	14.527	$\emptyset$	1	$\emptyset$	-6.58	1.87	134.35	$\emptyset$	$\emptyset$
	P3	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$
	P2	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$
	P1	$\emptyset$	$\emptyset$	$\emptyset$	$\emptyset$	-1	$\emptyset$	$\emptyset$	$\emptyset$	-1

## GOAL PROGRAMMING SOLUTION

## 0th Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	833.340	44.93	40.48	0	1	0	0	0	0
1 P2	d2-	624.520	32.25	34.38	1	0	0	0	0	0
1 P3	d3-	19.000	1	1	0	0	1	0	0	-1
	P3	19.000	1	1	0	0	0	0	0	-1
	P2	624.520	32.25	34.38	1	-1	0	0	0	0
	P1	833.340	44.93	40.48	-1	1	0	0	0	-1

## 1st Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x1	18.548	1	0.90	0	0.02	0	0	0	0
1 P2	d2-	26.362	0	5.32	1	-0.72	0	0	0	0
1 P3	d3-	0.452	0	0.10	0	-0.02	1	0	0	-1
	P3	0.452	0	0.10	0	-0.02	0	0	0	-1
	P2	26.362	0	5.32	1	-1.72	0	0	0	0
	P1	0.000	0	0	-1	0	0	0	0	-1

## 2nd Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x1	14.431	1	0	0	0.22	-9.10	0	0	9.10
1 P2	d2-	2.039	0	0	1	0.48	-53.76	0	0	53.76
$\emptyset -$	.x2	4.569	0	1	0	-0.22	10.10	0	0	-10.10
	P3	0	0	0	0	0	-1	0	0	0
	P2	2.039	0	0	1	-0.52	-53.76	0	0	53.76
	P1	0	0	0	-1	0	0	0	0	-1





## GOAL PROGRAMMING SOLUTION

## 0th Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	1021.840	76.24	57.35	0	1	0	0	0	0
1 P2	d2-	594.020	42.86	35.28	1	0	0	0	0	0
1 P3	d3-	15.000	1	1	0	0	1	0	0	-1
	P3	15.000	1	1	0	0	0	0	0	-1
	P2	594.020	42.86	35.28	1	-1	0	0	0	0
	P1	1021.840	76.24	57.35	-1	1	0	0	0	-1

## 1st Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x1	13.403	1	0.75	0	0.01	0	0	0	0
1 P2	d2-	19.570	0	3.04	1	-0.56	0	0	0	0
1 P3	d3-	1.597	0	0.25	0	-0.01	1	0	0	-1
	P3	1.597	0	0.25	0	-0.01	0	0	0	-1
	P2	19.570	0	3.04	1	-1.56	0	0	0	0
	P1	0	0	0	-1	0	0	0	0	-1

## 2nd Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x1	8.560	1	0	-0.25	0.15	0	0	0	0
$\emptyset -$	.x2	6.439	0	1	0.33	-0.18	0	0	0	0
1 P3	d3-	0.002	0	0	-0.08	0.03	1	0	0	-1
	P3	0.002	0	0	-0.08	0.03	0	0	0	-1
	P2	0	0	0	0	-1	0	0	0	0
	P1	0	0	0	-1	0	0	0	0	-1

## GOAL PROGRAMMING SOLUTION

0th Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	541.070	98.92	76.15	0	1	0	0	0	0
1 P2	d2-	276.030	51.76	36.76	1	0	0	0	0	0
1 P3	d3-	6.000	1	1	0	0	1	0	0	-1
	P3	6.000	1	1	0	0	0	0	0	-1
	P2	276.030	51.76	36.76	1	-1	0	0	0	0
	P1	541.070	98.92	76.15	-1	1	0	0	0	-1

1st Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
1 P1	d1-	13.541	0	5.90	-1.91	1	0	0	0	0
$\emptyset -$	.x1	5.333	1	0.71	0.02	0	0	0	0	0
1 P3	d3-	0.667	0	0.29	-0.02	0	1	0	0	-1
	P3	0.667	0	0.29	-0.02	0	0	0	0	-1
	P2	0	0	0	0	-1	0	0	0	0
	P1	13.541	0	5.90	-2.91	1	0	0	0	-1

2nd Iteration

		$\emptyset -$	$\emptyset -$	1 P1	1 P2	1 P3	$\emptyset -$	$\emptyset -$	1 P1	
		.x1	.x2	d1-	d2-	d3-	d1+	d2+	d3+	
$\emptyset -$	.x2	2.296	0	1	-0.32	0.17	0	0	0	0
$\emptyset -$	.x1	3.702	1	0	0.25	-0.12	0	0	0	0
1 P3	d3-	0.002	0	0	0.07	-0.05	1	0	0	-1
	P3	0.002	0	0	0.07	-0.05	0	0	0	-1
	P2	0	0	0	0	-1	0	0	0	0
	P1	0	0	0	-1	0	0	0	0	-1

Hence, the optimal solution obtained is --

$$x_1 = 19$$

$$x_2 = 15$$

$$x_3 = 6$$

So in order to maximize occupancy, revenue the Kolhapur  
✓ Municipal Transport should run 19 buses to sector 1 i.e.  
Bhawani Mandap and Shivaji Putla, 15 buses to sector 2 i.e.  
Gangavesh and Shahu Maidan and 6 buses to sector 3 i.e.  
Maharana Pratap Chowk and Sonya Maruti Chowk.

Similarly executing the program for three sectors we  
get bus allocation as follows --

i) Bhawani Mandap	:- 14 buses
ii) Shivaji Putla	:- 05 buses
iii) Gangavesh	:- 09 buses
iv) Shahu Maidan	:- 06 buses
v) Maharana Pratap Chowk	:- 04 buses
vi) Sonya Maruti Chowk	:- 02 buses

### 5.5 Analysis :-

After feeding the data as on 1st April, 1990 of Kolhapur Municipal Transport to the computer and after running the program for the solution by Goal Programming it is found that the total available buses which were forty in number should be distributed to the three sectors as follows --

i) Bhawani Mandap and Shivaji Putla	:- 19 buses
ii) Gangavesh and Shahu Maidan	:- 15 buses
iii) Maharana Pratap Chowk and Sonya Maruti Chowk	:- 06 buses
Total	:- 40 buses ✓

Further allocation/ distribution of buses within the sector, for each Control Point is obtained by running the Goal Program again for the data available separately for each Control Point and the buses allocated by goal program for consolidate/ combined data on the basis of route length. Thus, the program was run again for three times for each sector and the distribution of buses for each Control Point obtained is as follows --

i) Bhawani Mandap	:- 13 buses
ii) Shivaji Putla	:- 06 buses
iii) Gangavesh	:- 09 buses
iv) Shahu Maidan	:- 06 buses
v) Maharana Pratap Chowk	:- 04 buses
vi) Sonya Maruti Chowk	:- 02 buses
Total	:- 40 buses ✓

where as the present allocation of buses followed by Kolhapur Municipal Transport is as follows --

i) Bhawani Mandap	:- 12 buses
ii) Shivaji Putla	:- 04 buses
iii) Gangavesh	:- 09 buses
iv) Shahu Maidan	:- 07 buses
v) Maharana Pratap Chowk	:- 05 buses
vi) Sonya Maruti Chowk	:- 03 buses
Total	:- 40 buses ✓

The comparative statement for the present allocation and the result provided by Goal Programming is as follows --

Control Point	Present Allocation of buses	Allocation by Goal Programming	Present Revenue Per Km.	Expected Revenue Per Km.
Bhawani Mandap	12	13	89.76	97.24
Shivaji Putla	04	06	27.00	40.50
Gangavesh	09	09	73.62	73.62 ✓
Shahu Maidan	07	06	47.74	40.92 — ?
Maharana Pratap Chowk	05	04	37.55	30.04 — ?
Sonya Maruti Chowk	03	02	18.06	12.04
Total :-	40 ✓	40 ✓	293.73	294.36

Thus, by Goal Programming we may achieve a profit of 0.63 paise per kilometer.

The comparative statement for the present data, for first priority to Revenue and next priority to Occupancy is as follows --

Control Point	Present Allocation of buses	Allocation by Goal Programming	Present Revenue Per Km.	Expected Revenue Per Km.
Bhawani Mandap	12	14	89.76	104.72
Shivaji Putla	04	05	27.00	33.75
Gangavesh	09	09	73.62	73.62
Shahu Maidan	07	06	47.74	40.92
Maharana Pratap Chowk	05	04	37.55	30.04
Sonya Maruti Chowk	03	02	18.06	12.04
Total :-	40	40	293.73	295.09

Thus, by Goal Programming we may achieve a profit of 1.36 paise per kilometer.

Thus Kolhapur Municipal Transport should redistribute the buses as per the above schedule which in turn provides good service and good return in terms of revenue earned.